
[bookmark: _Toc411359862]Selecting nodes

[bookmark: _Toc411359863]Selects all anaphora within their tags:
for $anaphora := //anaph
return
$anaphora

[bookmark: _Toc411359864]Selects all singular nouns within their tags:
for $nouns in //w[@c5='NN1']
return
$nouns

[bookmark: _Toc411359865]Selects all nouns irrespective of number within their tags:
for $nouns in //w[starts-with(@c5,'NN')]
return
$nouns

[bookmark: _Toc411359866]Selects text of first colons in tricola:
for $tricola in //isoco[@ctype='tri']
return
string-join($tricola[@sequ='first']//w/text(),' ')
Note: instead of text(), string() can be used
[bookmark: _Toc411359867]Selects all sentences that do not host any <isoco> element:
for $sentences in //s[not(descendant::isoco)]
return
$sentences
Note: negative restriction is achieved by not() function
[bookmark: _Toc411359868]Selects all sentences that do not host either<isoco> or <anaph> elements:
for $sentences in //s[not(descendant::isoco) and not(descendant::anaph)]
return
$sentences

[bookmark: _Toc411359869]Returns the 'c5' values in each sentence:
for $s in //s
return
string-join($s//w[@c5]/string(@c5),' ')


[bookmark: _Toc411359870]Selects all words (<w> elements) in those sentences that do not host either<isoco> or <anaph> elements:
for $sentences in //s[not(descendant::isoco) and not(descendant::anaph)]
let $words := $sentences//w[not(c5='PUN')]
return
$words

[bookmark: _Toc411359871]Selects all sentences with less than (<) 5 words (excluding from the count instances of genitive –s tagged as a word)
for $sentences in //s[count(descendant::w[not(c5='PUN')]) < 5]
return
$sentences

[bookmark: _Toc411359872]Selects those sentences that host exactly 8 <isoco> elements:
for $sentences in //s[count(descendant::isoco) = 8]
return
$sentences

[bookmark: _Toc411359873]Using the count() function

[bookmark: _Toc411359874]Counts number of sentences:
for $whole_doc in //wtext
let $n_sentences := count($whole_doc//s)
return
$n_sentences

[bookmark: _Toc411359875]Counts number of sentences per paragraph (<p>) element:
for $paragraphs in //p
return
count($paragraphs//s)

[bookmark: _Toc411359876]Counts number of words:
for $whole_doc in //wtext
let $n_words := count($whole_doc//w)
return
$n_words

[bookmark: _Toc411359877]Counts words per paragraph (excluding instances of genitive –s tagged as <w>):
for $paragraphs in //p
return
count($paragraphs//w[not(@c5="PUN")

[bookmark: _Toc411359878]Counts genitive-s tagged as a word (e.g., <w id="15.20" c5="POS">'s</w>):
for $words in //wtext
let $s := count($words//w[@c5="POS"])
return
$s

[bookmark: _Toc411359879]Counts number of anaphora:
for $whole_doc in //wtext
let $n_anaphora := count($whole_doc//anaph)
return
$n_anaphora

[bookmark: _Toc411359880]Counts number of isocola:
for $whole_doc in //wtext
let $n_isocola := count($whole_doc//isoco)
return
$n_isocola

[bookmark: _Toc411359881]Counts number of words in all anaphora:
for $whole_doc in //wtext
let $n_words_anaphora := count($whole_doc//anaph//w)
return
$n_words_anaphora

[bookmark: _Toc411359882]Counts number of words in all isocola:
for $whole_doc in //wtext
let $n_words_isocola := count($whole_doc//isoco//w)
return
$n_words_isocola

[bookmark: _Toc411359883]Counts number of words for each component of anaphora:
for $anaphora in //anaph
let $w := count($anaphora//w)
return
$w
Note: no restrictions to the context node are necessary because <anaph>elements, in this corpus, do not nest in one another.

[bookmark: _Toc411359884]Counts number of <isoco> elements per sentence:
for $sentences in //s
let $n_isocola := count($sentences//isoco)
return
$n_isocola
[bookmark: _Toc411359885]Counts number of <isoco> and <anaph> elements per sentence:
for $sentences in //s
let $n_isocola := count($sentences//isoco)
let $n_anaphora := count($sentences//anaph)
return
concat($n_isocola,';', $n_anaphora)

[bookmark: _Toc411359886]Counts number of words per colon for dicola:
for $isocola in //isoco[@ctype='di']
let $words_first.colon := count($isocola[@sequ='first']//w[not(@c5="POS")])
let $words_second.colon := count($isocola[@sequ='second']//w[not(@c5="POS")])
let $sequ_value := $isocola[@sequ]
return
concat($isocola[@sequ]/string(@sequ),';',$words_first.colon,';',$words_second.colon)

[bookmark: _Toc411359887]Counts number of words per colon for tricola:
for $isocola in //isoco[@ctype='tri']
let $words_first.colon := count($isocola[@sequ='first']//w[not(@c5="POS")])
let $words_second.colon := count($isocola[@sequ='second']//w[not(@c5="POS")])
let $words_third.colon := count($isocola[@sequ='third']//w[not(@c5="POS")])
let $sequ_value := $isocola[@sequ]
return
concat($isocola[@sequ]/string(@sequ),';',$words_first.colon,';',$words_second.colon,';',$words_third.colon)

[bookmark: _Toc411359888]Counts number of words per colon for tetracola:
for $isocola in //isoco[@ctype='tet']
let $words_first.colon := count($isocola[@sequ='first']//w[not(@c5="POS")])
let $words_second.colon := count($isocola[@sequ='second']//w[not(@c5="POS")])
let $words_third.colon := count($isocola[@sequ='third']//w[not(@c5="POS")])
let $words_fourth.colon := count($isocola[@sequ='fourth']//w[not(@c5="POS")])
let $sequ_value := $isocola[@sequ]
return
concat($isocola[@sequ]/string(@sequ),';',$words_first.colon,';',$words_second.colon,';',$words_third.colon,';',$words_fourth.colon)


[bookmark: _Toc411359889]Traversing axes

[bookmark: _Toc411359890]Counts the number of words in all anaphora that are children to <isoco> elements:
for $whole_text in //wtext
let $n_words := count($whole_text//anaph[parent::isoco]//w[not(@c5="POS")])
return
$n_words

[bookmark: _Toc411359891]Selects all isocola that host an <isoco> child or <isoco> descendant:
for $isocola in //isoco
let $isoco_descendants := $isocola[child::isoco]
return
$isoco_descendants

[bookmark: _Toc411359892]Selects all isocola whose ancestor is an <isoco> element:
for $isocola in //isoco
let $isoco_descendants := $isocola[ancestor::isoco]
return
$isoco_descendants

[bookmark: _Toc411359893]Selects all anaphora which are contained within (i.e. descendants to) isocola:
for $anaphora in //anaph
let $anaph_in_isoco := $anaphora[ancestor::isoco]
return
$anaph_in_isoco

[bookmark: _Toc411359894]Selects all anaphora which are not contained within (i.e. descendants to) isocola:
for $anaphora in //anaph
let $anaph_out_isoco := $anaphora[not(ancestor::isoco)]
return
$anaph_out_isoco

[bookmark: _Toc411359895]Selects those possessive determiners that are immediately followed by a noun (singular or plural):
for $dps in //w[@c5='DPS' and following::w[1][starts-with(@c5,'NN')]]
return
$dps



[bookmark: _Toc411359896]Selects those nouns (irrespective of number) that are immediately preceded by a possessive determiner:
for $nn in //w[starts-with(@c5,'NN') and preceding::w[1][@c5='DPS']]
return
$nn

[bookmark: _Toc411359897]Returns those singular nouns immediately following a possessive determiner:
for $dps in //w[@c5='DPS']
let $nn := $dps/following-sibling::w[1][@c5='NN1']
return
$nn


[bookmark: _Toc411359898]Using the position() and last() functions

[bookmark: _Toc411359899]Returns the first word at every axis:
for $first_word in //w[position()=1]
return
$first_word

[bookmark: _Toc411359900]Returns the last word at every axis:
for $last_word in //w[last()]
return
$last_word

[bookmark: _Toc411359901]Returns the last word in every last <isoco> element in the sentence:
for $s in //s
let $last_word := $s//descendant::isoco[last()]//w[last()]
return
$last_word

[bookmark: _Toc411359902]Returns the last word in each sentence:
for $s in //s
let $last_word_sentence := $s/descendant::w[last()]
return
$last_word_sentence/string()

[bookmark: _Toc411359903]Returns all nouns (irrespective of number) that occur in a window of 1-4 words after a possessive determiner:
for $dps in //w[@c5='DPS']
let $nn := $dps/following-sibling::w[position()=1 to 4][starts-with(@c5,'NN')]
return
$nn



[bookmark: _Toc411359904]Advanced queries
[bookmark: _Toc411359905]Computing collocates
[bookmark: _Toc411359906]Returns all words occurring after the word “our” in a 2-word window on the right 
for $our in //w[text()='our']
let $w_slot1 :=  $our/following::w[1] 
let $w_slot2 :=  $our/following::w[2] 
return
concat($our[@id]/string(@id),'; ',$our/string(),' ',$w_slot1/string(),' ',$w_slot2/string())
Note: using text() it is possible to search for text independently of its POS value

[bookmark: _Toc411359907]Returns all nouns (irrespective of number) that occur in a window of 1 to 3 words after a possessive determiner:
for $dps in //w[@c5='DPS']
let $nn_pos1 :=  $dps/following::w[position()=1][starts-with(@c5,'NN')] 
let $nn_pos2 :=  $dps/following::w[position()=2][starts-with(@c5,'NN')] 
let $nn_pos3 :=  $dps/following::w[position()=3][starts-with(@c5,'NN')] 
return
concat($dps[@id]/string(@id),';',$dps/string(),';',$nn_pos1/string(),';',$nn_pos2/string(),';',$nn_pos3/string())


[bookmark: _Toc411359908]Returns all verb forms (regardless of verb type) occurring in a window of 1 to 3 words after a wh-word (in CLAWS5 wh-words can be tagged as 'PNQ' (e.g., who), 'AVQ' (e.g., why, when), and 'DTQ' (e.g., what, which, whose):
for $wh in //w[ends-with(@c5,'Q')]
let $v_pos1 :=  $wh/following::w[position()=1][starts-with(@c5,'V')] 
let $v_pos2 :=  $wh/following::w[position()=2][starts-with(@c5,'V')] 
let $v_pos3 :=  $wh/following::w[position()=3][starts-with(@c5,'V')] 
return
concat($wh[@id]/string(@id),';',$wh/string(),';',$v_pos1/string(),';',$v_pos2/string(),';',$v_pos3/string())




[bookmark: _Toc411359909]Returns all adjectives occurring in a window of 1 to 3 words before a noun (regardless of number and type) occurring in a window of 1 to 3 words after a wh-word (in CLAWS5 wh-words can be tagged as 'PNQ' (e.g., who), 'AVQ' (e.g., why, when), and 'DTQ' (e.g., what, which, whose):
for $n in //w[starts-with(@c5,'N')]
let $adj_pos-1 :=  $n/preceding::w[position()=1][@c5='AJ0']
let $adj_pos-2 :=  $n/preceding::w[position()=2][@c5='AJ0']
let $adj_pos-3 :=  $n/preceding::w[position()=3][@c5='AJ0']
return
concat($n[@id]/string(@id),';',$n/string(),';',$adj_pos-1/string(),';',$adj_pos-2/string(),';',$adj_pos-3/string())

[bookmark: _Toc411359910]Returns all verb forms (of any type) occurring in a window of 1 to 3 words before an adverb particle (e.g., up, out):
for $avp in //w[@c5='AVP']
let $v_pos1 :=  $avp/preceding::w[1][starts-with(@c5,'V')]
let $v_pos2 :=  $avp/preceding::w[2][starts-with(@c5,'V')]
let $v_pos3 :=  $avp/preceding::w[3][starts-with(@c5,'V')]
return
concat($v_pos3/string(),';',$v_pos2/string(),';',$v_pos1/string(),';',$avp/string(),';',$avp[@id]/string(@id))

[bookmark: _Toc411359911]Returns all words occurring in a window of 1 to 2 words left and right of a verb form (of any type):
for $v in //w[starts-with(@c5,'V')]
let $w_left1 :=  $v/preceding::w[1]
let $w_left2 :=  $v/preceding::w[2]
let $w_right1 :=  $v/following::w[1]
let $w_right2 :=  $v/following::w[2]
return
concat($w_left2/string(),';',$w_left1/string(),';',$v[@id]/string(@id),';',$v/string(),';',$w_right1/string(),';',$w_right2/string())

[bookmark: _Toc411359912]Returns all adjectives occurring in a window of 1 to 2 words left and right of a noun (of any number and type):
for $n in //w[starts-with(@c5,'N')]
let $adj_left1 :=  $n/preceding::w[1][@c5='AJ0']
let $adj_left2 :=  $n/preceding::w[2][@c5='AJ0']
let $adj_right1 :=  $n/following::w[1][@c5='AJ0']
let $adj_right2 :=  $n/following::w[2][@c5='AJ0']
return
concat($adj_left2/string(),';',$adj_left1/string(),';',$n[@id]/string(@id),';',$n/string(),';',$adj_right1/string(),';',$adj_right2/string())


[bookmark: _Toc411359913]Computing n-grams

[bookmark: _Toc411359914]Produces an ordered list of all 3-grams:
for $ngram_pos1 at $sequ in //w
let $ngram_pos2 :=  $ngram_pos1/following::w[position()=1]
let $ngram_pos3 :=  $ngram_pos1/following::w[position()=2]
return
concat($sequ,'. ', $ngram_pos1/string(),' ', $ngram_pos2/string(),' ', $ngram_pos3/string())
Note: the sequence numbers are added by means of the keyword at followed by a variable name for sequence (here: at $sequ)

[bookmark: _Toc411359915]Returns all 4-POS-grams:
for $ngram_pos1 in //w[not(@c5='POS')]
let $ngram_pos2 :=  $ngram_pos1/following::w[position()=1]
let $ngram_pos3 :=  $ngram_pos1/following::w[position()=2]
let $ngram_pos4 :=  $ngram_pos1/following::w[position()=3]
return
concat($ngram_pos1[@id]/string(@id),';',$ngram_pos1[@c5]/string(@c5),';',$ngram_pos2[@c5]/string(@c5),';',$ngram_pos3[@c5]/string(@c5),';',$ngram_pos4[@c5]/string(@c5))



[bookmark: _Toc411359916]Constructing frequency lists

[bookmark: _Toc411359917]Constructs a list of word types in descending order of frequency:
let $tokens :=
for $w in //w[not(@c5='POS')]
	return
	lower-case($w/string()) 
let $types := distinct-values($tokens)
for $type_list in $types
let $freq := count($tokens[.=$type_list])
where $freq >= 5
order by $freq descending
return 
concat($type_list,';',$freq)

[bookmark: _Toc411359918]Constructs a list of POS tags in ascending order of frequency:
let $pos_tokens:=
for $w in //w[@c5]
 	return
	$w[@c5]/string(@c5)
let $types := distinct-values($pos_tokens)
for $type_list in $types
let $freq := count($pos_tokens[.=$type_list])
where $freq >= 5
order by $freq ascending
return 
concat($type_list,';',$freq)

[bookmark: _Toc411359919]Constructs a list of word token bigrams in descending order of frequency:
let $tokens := 
    for $w in //w
    return
        concat(lower-case($w/string()),' ',lower-case($w/following::w[1]/string()))     
let $types := distinct-values($tokens)
for $type in $types
let $freq := count($tokens[.=$type])
where $freq >= 3
order by $freq descending
return
    concat($type,';', $freq)



[bookmark: _Toc411359920]Constructs a list of POS trigrams in descending order of frequency:
let $pos_tokens:=
for $w in //w[@c5]
 	return
	concat($w[@c5]/string(@c5),' ',$w[@c5]/following::w[1]/string(@c5),' ',$w[@c5]/following::w[2]/string(@c5))
let $types := distinct-values($pos_tokens)
for $type_list in $types
let $freq := count($pos_tokens[.=$type_list])
where $freq >= 5
order by $freq descending
return 
concat($type_list,';',$freq)




[bookmark: _Toc411359921]Calculating positions

[bookmark: _Toc411359922]Calculates positions of possessive pronouns (DPS):
for $whole_doc in //wtext
let $words_total := count($whole_doc//w[not(@c5="POS")])
for $dps in //w[@c5='DPS']
let $words_prec_dps := count($dps/preceding::w[ancestor::wtext])
let $pos := $words_prec_dps div $words_total
return
concat($pos,';',$dps/text())

[bookmark: _Toc411359923]Calculates positions of anaphora; also returns corresponding values on 'repeat' attributes:
for $whole_doc in //wtext
let $words_total := count($whole_doc//w[not(@c5="POS")])
for $anaph in //anaph
let $words_prec := count($anaph/preceding::w[ancestor::wtext])
let $pos := $words_prec div $words_total
return
concat($anaph[@repeat]/string(@repeat),';',$pos)

[bookmark: _Toc411359924]Calculates positions of first colons in tricolons: 
for $whole_doc in //wtext
let $words_total := count($whole_doc//w[not(@c5="POS")])
for $first.colon in //isoco[@ctype='tri'][@sequ='first']
let $words_colon := count($first.colon//w[not(@c5="POS")])
let $words_prec_first.colon := count($first.colon/preceding::w[ancestor::wtext])
let $pos := $words_prec_first.colon div $words_total
return
concat($words_colon,';',$pos)

[bookmark: _Toc411359925]Calculates positions of last colons:
for $whole_doc in //wtext
let $words_total := count($whole_doc//w[not(@c5="POS")])
for $last.colon in //isoco[@ctype='di'][@sequ='second']
let $words_colon := count($last.colon//w[not(@c5="POS")])
let $words_prec_last.colon := count($last.colon/preceding::w[ancestor::wtext])
let $pos := $words_prec_last.colon div $words_total
return
concat($words_colon,';',$pos)
Variable values in line 3: @ctype: 'di', 'tri', 'tet'; @sequ: 'second', 'third', 'fourth'



[bookmark: _Toc411359926]Calculates positions of the last word in the last colon; also returns that last word and its 'c5' value:
for $whole_doc in //wtext
let $words_total := count($whole_doc//w[not(@c5="POS")])
for $last_colon in //isoco[@ctype='tet'][@sequ='fourth']
let $last_word_colon := $last_colon/w[last()]
let $words_prec_last_word_colon := count($last_word_colon/preceding::w[ancestor::wtext])
let $pos := $words_prec_last_word_colon div $words_total
return
concat($last_word_colon/string(),';',$last_word_colon[@c5]/string(@c5),';',$pos)
Variable values in line 3: @ctype: 'di', 'tri', 'tet'; @sequ: 'second', 'third', 'fourth'


[bookmark: _Toc411359927]Constructing Key Word In Context (KWIC) displays

[bookmark: _Toc411359928]Constructs a KWIC display for the search word “nation” in a 5-word window left and right of the node (L5-R5)
let $search_word := 'our'
let $hits := //w[.=$search_word]
let $window_size := 5
for $hit at $sequ in $hits
let $left_context := 
    for $slot in reverse(1 to $window_size)
    let $left_w := $hit/preceding::w[$slot]
    return 
        $left_w
let $right_context := 
    for $slot in (1 to $window_size)
    let $rw := $hit/following::w[$slot]
	return
        $rw
return
    concat($sequ,'. ',string-join($left_context/string(),' '), ' |', $hit/string(), '| ', string-join($right_context/string(),' '))



[bookmark: _Toc411359929]Constructs a L3-R3 KWIC output for all adjectives (tagged 'AJ0') and all surrounding POS values:
let $search_pos := //w[@c5='AJ0']
let $hits := //w[.=$search_pos]
let $window_size := 3
for $hit at $sequ in $hits
let $left_context := 
    for $slot in reverse(1 to $window_size)
    let $left_w := $hit/preceding::w[$slot]
    return 
        $left_w
let $right_context := 
    for $slot in (1 to $window_size)
    let $rw := $hit/following::w[$slot]
	return
        $rw
return
    concat($sequ,'. ',string-join($left_context[@c5]/string(@c5),' '), ' |', $hit/string(), '| ', string-join($right_context[@c5]/string(@c5),' '))

First 4 lines of output:
1. DPS |fellow| NN2 PNP VVB
2. NN1 PRP PNP |grateful| PRP AT0 NN1
3. PNP VHB VVN |mindful| PRF AT0 NN2
4. AV0 VVN AT0 |presidential| NN1 AT0 NN2
